ASYMMETRIC-SPECTRA METHODS FOR ADAPTIVE FM SYNTHESIS

Victor Lazzarini, Joseph Timoney and Thomas Lysaght

Sound and Digital Music Technology Group
National University of Ireland, Maynooth
Ireland
VLazzarini@nuim.ie,
JTimoney@cs.nuim.ie,
TLysaght@cs.nuim.ie

ABSTRACT

This article provides an overview of further methods for producing hybrid natural-synthetic spectra with adaptive frequency modulation (AdFM). It focuses on three different techniques for the generation of asymmetric spectra based on single-sideband FM, asymmetric FM and Split-sideband synthesis. The first two techniques are applied to the variable delay line implementation of AdFM, whereas the third is based on an extension of the heterodyne method. The article discusses the principles involved in each synthesis technique in good detail, providing one reference implementation for each. A number of examples are discussed, demonstrating the possibilities for a variety of digital audio effects applications.

1. INTRODUCTION

Adaptive Frequency Modulation (AdFM)[1][2] has been proposed as a novel way to generate hybrid natural-synthetic tones for musical use. It builds on the principles of frequency (or phase) modulation synthesis, as first outlined for musical signals by Chowning[3]. By substituting the synthetic sinusoidal carrier for an arbitrary audio signal, we allow it to become an adaptive effect[4]. The technique can use two basic methods of phase modulation, through the use of a variable delay line (fig.1) or by heterodyning (fig.2), using a rearrangement of the FM formula (see [1] and [2] for further details on these methods). In order to allow for various carrier to modulator ratios, the fundamental frequency of the carrier signal is tracked using a standard estimation algorithm.

The resulting spectra of AdFM instruments will depend on two basic factors: the type of input used and the FM parameters employed. As with standard FM synthesis, the spread of energy for a given spectrum is dependent on the index of modulation and the sidebands are given by Bessel functions of the first kind of increasing orders. In general, the spectrum will be symmetric around the carrier frequency. In the case of AdFM, each component of the input spectrum will count as a carrier, around which symmetric components will be placed as a result of the process.

In this article, we propose three methods for the modification of the symmetry in the energy spread of AdFM components. The first method is based on a summation formulae[5]-inspired variation to FM synthesis, first proposed by Moorer[6], which generates single-sideband spectra. The next technique is both a generalisation and an improvement to this technique, developed by Palamin et al[7]. Using a similar approach of amplitude-modulating the FM output, it provides an extremely flexible way of generating various asymmetric sideband outputs.



Figure 1. Delay-based AdFM

Finally, we will examine an original method, proposed by these authors, based on an adaptive version of Split-Sideband Synthesis (SpSB)[8]. Instead of trying to obtain direct ways of modifying sideband amplitudes, as in the previous two techniques, we use a combination of single-sideband modulation[9] and the heterodyne method of FM synthesis. This provides a flexible way of splitting the sidebands into four separate groups (upper even, upper odd, lower even and lower odd). The method, first proposed as a novel means of distortion synthesis, proves to be very well suited to adaptive applications.

2. ASYMMETRIC DELAY-BASED ADFM

The first two techniques of sideband modification will be applied to the variable delay line phase modulation algorithm. Since this method is effectively an alternative implementation of the original FM formula, these techniques translate well to AdFM, if due care with modulator phase offsets is taken.



Figure 2. Heterodyne AdFM

2.1. Single-sideband AdFM

Single-sideband spectra can be produced by ring-modulating the output of simple FM with an exponential mapping of a cosine signal. This demonstrated by the following expressions:

$$s_{upper}(t) = e^{k\cos(\omega_m t)-k}\sin(\omega_c t + k\sin(\omega_m t)) = \\ = \frac{1}{e^k}\sum_{n=0}^{\infty}\frac{k^n}{n!}\sin(\omega_c t + n\omega_m t) \quad (1)$$

$$s_{lower}(t) = e^{k\cos(\omega_m t)-k}\sin(\omega_c t - k\sin(\omega_m t)) = \\ = \frac{1}{e^k}\sum_{n=0}^{\infty}\frac{k^n}{n!}\sin(\omega_c t - n\omega_m t) \quad (2)$$

The output will contain the upper (or lower) sidebands, made up of the sums (or differences) of the carrier frequency and integer multiples of the modulation frequency. However, unlike the original FM, the scaling of each component will not depend on Bessel functions, but directly on the values of the index of modulation (k in Eq.1), raised to increasingly higher powers. This will produce quite a different spectral envelope from original FM.

The single-sideband expression in Eqs.1-2 can be applied to the delay-based AdFM (Fig.1) simply by ring modulating its output. However, in order to produce the desired effect of sideband cancellation, it is important that phases are correctly set. Thankfully, carrier phases are not relevant here, as these would be very difficult to estimate and control for arbitrary signals. We will be concerned only with matching the two modulator phases.

In order to match the phase offsets in Eq.1, the variable delay line modulator needs to be given an offset of $\pi/2$ radians (relative to cosine phase). In addition, this sinusoid needs to be raised and scaled for use in delay line modulation (DC offset of 1 and scaling by 0.5). The exponentially-mapped sinusoid should of course be in cosine phase. To realise Eq.2, we can either invert the modulator wave phase or change the sign of its frequency.

A reference implementation of single-sideband AdFM, in Csound5[10], is shown below:

```
/* SSDFM opcode
aout SSDFM asig,krat,kndx,ifn

asig - input
krat - c:m ratio
knx - index of modulation
ifn – delay mod func table
(raised, scaled, inverted sine)
ifn2 –  cosine func table
*/

opcode SSDFM,a,akkii
 setksmps 1
 as,kfm,knx,ifn,ifn2 xin
 /* pitch tracking */
 kcps,kamp ptrack as, 1024
 kcps port kcps, 0.01
 /* modulator */
 adt oscili  knx/($M_PI*kcps),kcps/kfm,ifn
 /* variable delay */
 adp delayr   .1
 adel  deltapx  adt+2/44100,2
       delayw  as
 /* ring mod signal */
 amod oscili knx, kcps/kfm, ifn2
       xout adel*exp(amod-knx)
endop
```

2.2. Asymmetric AdFM

An extension and generalisation of the principles embodied the single-sideband formula was proposed by J.P. Palamin and others in their 1988 work on asymmetric FM synthesis. They propose the introduction of an extra parameter, r in the equation below, which can control spectral symmetry (with I_0 denoting the modified Bessel function of the first kind of order 0 and J_n the Bessel function of the first kind of order n):

$$s(t) = e^{\frac{k}{2}(r-\frac{1}{r})\cos(\omega_m t)-\frac{1}{2}\ln\left[I_0(k[r-\frac{1}{r}])\right]} \times \\ \sin(\omega_c t + \frac{k}{2}(r+\frac{1}{r})\sin(\omega_m t)) = \\ = \frac{1}{\sqrt{I_0(k[r-1/r])}}\sum_{n=-\infty}^{\infty} r^n J_n(k)\sin(\omega_c t + n\omega_m t) \quad (3)$$

This formula preserves the scaling by Bessel functions, but introduces a means of dislocating the centre of symmetry in the FM spectrum. Furthermore, it allows a transition from the original FM spectrum ($r = 1$), to modified ones, where the centre of symmetry can be in the upper sidebands ($r > 1$) or in the lower ($r < 1$).

A variation of the above algorithm is also possible, whereby if we invert the blocks (*r-1/r*) and (*r+1/r*), we will get the spectrum scaled by modified Bessel functions[10] of different orders (I_n):

$$s(t) = e^{\frac{k}{2}(r+\frac{1}{r})\cos(\omega_m t) - \frac{1}{2}\ln\left[I_0(k[r+\frac{1}{r}])\right]} \times \sin(\omega_c t + \frac{k}{2}(r-\frac{1}{r})\sin(\omega_m t)) = \frac{1}{\sqrt{I_0(k[r+1/r])}}\sum_{n=-\infty}^{\infty} r^n I_n(k)\sin(\omega_c t + n\omega_m t) \qquad (4)$$

One of the significant results of Eq.4 is that now, with r=1 (and the symmetric spectrum around the carrier) higher-order sidebands will never be of larger magnitude than lower-order ones (that is, sideband n will always have less energy than n-1). Figure 3 shows a plot of the scaling functions $I_n(k)I_0(2k)^{-1/2}$ for each sideband n in Eq.4 against the index of modulation k, with r=1. This is a direct consequence of the appearance of the modified Bessel functions in the formula, as these are non-oscillating and increase exponentially. Note that this feature is lost if r is not 1, as the spectrum will be asymmetric. Nevertheless, this algorithm will allow for the generation of undistorted 'moving formants'.

Again, the application of the principles of Eqs. 3 and 4 to AdFM is quite straightforward. With the phase considerations of the previous method in mind, we now only require the use of the expression that includes the r parameter. We will also require, for normalisation purposes, the logarithm of the modified Bessel function of order 0 (I_0 in Eq.3), which can be obtained by table lookup.

A reference implementation of asymmetric AdFM, in Csound 5, is shown below:

```
/* ASDFM1 opcode
aout ASDFM1 asig,krat,kndx,if1,if2,if3,imx

asig - input
krat - c:m ratio
knx - index of modulation
kR – symmetry control
if1 – delay mod func table
(raised, scaled, inverted sine)
if2 –  cosine func table
if3 -  ln of mod Bessel function
imx – max value of knx(kR – 1/kR)
*/

opcode ASDFM1,a,akkkiiii
 setksmps 1
 as,kfm,knx,kR,ifn,ifn2,ifn3,imax xin
 /* pitch tracking */
 kcps,kamp ptrack as, 1024
 kcps port kcps, 0.01
 /* delay modulation */
 kndx = (knx/2)*(kR+1/kR)
 adt oscili  kndx/($M_PI*kcps),kcps/kfm,ifn
 /* variable delay line */
 adp  delayr   .1
 adel deltapx  adt+2/44100,2
      delayw  as
 /* ring modulation signal */
 amod oscili 1, kcps/kfm, ifn2
 kndx2 = knx*(kR-1/kR)
 kln tablei kndx2/imax, ifn3, 1
    xout adel*exp(.5*(kndx2*amod - kln))

endop
```



Figure 3. Scaling functions $I_n(k)I_0(2k)^{-1/2}$ for each sideband n Eq.4, with r=1, plotted against the index of modulation k.

```
/* ASDFM2 opcode
aout ASDFM2 asig,krat,kndx,if1,if2,if3,imx

asig - input
krat - c:m ratio
knx - index of modulation
kR – symmetry control
if1 – delay mod func table
(raised, scaled, inverted sine)
if2 –  cosine func table
if3 -  ln of mod Bessel function
imx – max value of knx(kR – 1/kR)
*/

opcode ASDFM2,a,akkkiiii
 setksmps 1
 as,kfm,knx,kR,ifn,ifn2,ifn3,imax xin
 /* pitch tracking */
 kcps,kamp ptrack as, 1024
 kcps port kcps, 0.01
 /* delay modulation */
 kndx = (knx/2)*(kR-1/kR)
 adt oscili  kndx/($M_PI*kcps),kcps/kfm,ifn
 /* variable delay line */
 adp  delayr   .1
 adel deltapx  adt+2/44100,2
      delayw  as
 /* ring modulation signal */
 amod oscili 1, kcps/kfm, ifn2
 kndx2 = knx*(kR+1/kR)
 kln tablei kndx2/imax, ifn3, 1
    xout adel*exp(.5*(kndx2*amod - kln))

endop
```

3. THE SPLIT-SIDEBAND METHOD

SpSB has been developed effectively as an extension of the heterodyne AdFM method. In essence, it is a non-linear distortion technique which has strong links with FM, Waveshaping[12] and Single-sideband modulation, but is novel in its formulation. The SpSB method uses familiar distortion synthesis parameters such as modulation/carrier frequencies and modulation index. It produces four independent outputs, containing the resulting complex spectra in separate sideband groups: lower-odd, lower-even, upper-odd and upper-even (fig.4). These signals can be then mixed down in a variety of combinations and at different levels to produce different spectra, or they can be further processed, spatialised, etc..



Figure 4. SpSB synthesis

SpSB synthesis is based on two main principles: (i) production of independent odd and even sidebands; and (ii) separation of lower and upper sideband groups. The first is realised by a combination of amplitude (ring) modulation and function mapping (similar to waveshaping) and the second is enabled by using complex analytic signals.

3.1. Generating split sidebands

The basic method for producing even and odd sideband components uses the following signals

$$s(t) = \cos(I \sin(\omega t + \phi)) \tag{5}$$

and

$$s(t) = \sin(I \sin(\omega t) + \phi) \tag{6}$$

which have the following expansions, respectively[13]:

$$s(t) = J_0(I) + 2\sum_{n=1}^{\infty} J_{2n}(I)\cos(2n\omega t + 2n\phi) \tag{7}$$

$$s(t) = 2\sum_{n=1}^{\infty} J_{2n-1}(I)\sin([2n-1]\omega t + [2n-1]\phi) \tag{8}$$

The resulting spectra in both cases will be harmonic, with the fundamental frequency in both cases being assigned to the frequency ω (missing in the spectrum of Eq.5, but implied by relationships in the higher harmonics). To centre the spectra at any frequency, we ring-modulate both signals with a sinewave carrier at ω_c:

$$s_{even}(t) = \sin(\omega_c t)\cos(I \sin(\omega_m t)) \tag{9}$$

and

$$s_{odd}(t) = \sin(\omega_c t)\sin(I \sin(\omega_m t)) \tag{10}$$

This method generates two sideband groups, separately: the even sidebands (Eq.9), $\omega_c \pm 2n\omega_m$; and the odd sidebands (Eq.10), $\omega_c \pm (2n-1)\omega_m$.

Separating these two sideband groups into further two, placed above and below the carrier frequency is a matter of modifying the ring modulation of Eqs. 9 and 10 slightly. Instead of using real signals, we will multiply analytic signals. These will contain either positive or negative frequencies only. In that case our complex sinewave carrier $x(t)$ will be defined as:

$$x(t) = \sin(\omega_c t) - j\cos(\omega_c t) \tag{11}$$

Due to peculiarities of their expansions (Eqs.7 and 8), it is not possible to produce similarly a synthetic quadrature signal for the cosine or sine-mapped signals of Eqs.5 and 6. To solve this, we will instead apply a Hilbert Transform[14] to produce the correct phase delays needed. We can define the Hilbert Transform $H\{x\}$ of an arbitrary signal as:

$$H\{s(t)\} = s(t) * \frac{1}{\pi t} = \frac{1}{\pi}\int_{-\infty}^{\infty} \frac{s(\tau)}{t-\tau}\partial\tau \tag{12}$$

This results in a constant phase shift of $\pi/2$ across the positive spectrum, thus generating the required quadrature signal. Consequently, we can produce signals exhibiting only positive and negative frequencies, respectively:

$$z_{pos}(t) = s(t) + jH\{s(t)\} \tag{13}$$

$$z_{neg}(t) = s(t) - jH\{s(t)\} \tag{14}$$

Now, by realising single-sideband modulation instead of the ring modulation in Eqs.9 and 10, we have Split-Sideband Synthesis. The upper and lower sidebands are produced by the following matricial heterodying of signals:

$$\begin{bmatrix} s_{upper,even}(t) & s_{upper,odd}(t) \\ s_{lower,even}(t) & s_{lower,odd}(t) \end{bmatrix} = \begin{bmatrix} \sin(\omega_c t) & \cos(\omega_c t) \\ \sin(\omega_c t) & -\cos(\omega_c t) \end{bmatrix} \times \begin{bmatrix} \cos(I\sin(\omega_m t)) & \sin(I\sin(\omega_m t)) \\ H\{\cos(I\sin(\omega_m t))\} & H\{\sin(I\sin(\omega_m t))\} \end{bmatrix} \tag{15}$$

The spectra of the four outputs is shown in the following expansions (Eqs.16-19):

$$s_{upper,even}(t) = J_0(I)\sin(\omega_c t) + 2\sum_{n=1}^{\infty} J_{2n}(I)\sin(\omega_c + 2n\omega_m t) \tag{16}$$

$$s_{upper,odd}(t) = 2\sum_{n=1}^{\infty} J_{2n-1}(I)\cos(\omega_c + [2n-1]\omega_m t) \tag{17}$$

$$s_{lower,even}(t) = J_0(I)\sin(\omega_c t) + 2\sum_{n=1}^{\infty} J_{2n}(I)\sin(\omega_c - 2n\omega_m t) \tag{18}$$

$$s_{lower,odd}(t) = 2\sum_{n=1}^{\infty} - J_{2n-1}(I)\cos(\omega_c - [2n-1]\omega_m t) \tag{19}$$



Figure 5. The ASpSB design

3.2. Adaptive Split-sideband Synthesis

Implementing the SpSB with arbitrary input signals is quite straightforward. Instead of employing the complex sinusoidal carrier of Eq.15 we will generate a quadrature signal using the Hilbert transform, as applied above to the modulator signal. The mixing matrix for Adaptive SpSB (ASpSB) is shown below in Eq.20

$$\begin{bmatrix} s_{upper,even}(t) & s_{upper,odd}(t) \\ s_{lower,even}(t) & s_{lower,odd}(t) \end{bmatrix} = \begin{bmatrix} x(t) & -H\{x(t)\} \\ x(t) & H\{x(t)\} \end{bmatrix} \times \begin{bmatrix} \cos(I\sin(\omega_m t)) & \sin(I\sin(\omega_m t)) \\ H\{\cos(I\sin(\omega_m t))\} & H\{\sin(I\sin(\omega_m t))\} \end{bmatrix} \tag{20}$$

As with AdFM, we will also pitch track the input signal in order to have control over the carrier to modulator ratio. A signal flowchart for ASpSB is shown on fig.5. The following Csound5 opcode serves as a reference implementation:

```
/* ASpSB opcode
a1,a2,a3,a4 ASpSB asig, krat, kndx, ifn

a1,a2,a3,a4 - upper/even, upper/odd,
       lower/even, lower/odd outputs
asig - input
krat - c:m ratio
kndx - index of modulation
ifn - sinewave function table number
*/

opcode ASpSB,aaaa,akki

 asig,krat,kndx,ifn  xin

 ; modulator signals
 kfc,kac ptrack asig, 512
 a1 oscili kndx/(2*$M_PI),kfc/krat,ifn
 a2 tablei a1,ifn,1,0.25,1
 a3 tablei a1,ifn,1,0,1

 ; complex modulators
 aae, abe hilbert a2; sidebands, analytic
 aao, abo hilbert a3; odd sbs, analytic

 ; complex carrier
 ac,ad hilbert asig

 ;  even and odd sidebands,
 ;  lower/upper sides
 aeu = aae*ac - abe*ad
 aou = aao*ac - abo*ad
 ael = aae*ac + abe*ad
 aol = aao*ac + abo*ad

     xout  aeu,aou,ael,aol
endop
```

4. DISCUSSION AND EXAMPLES

The three different methods have distinct timbral features, which make them interesting alternatives to the original AdFM processes.

4.1. Single-sideband AdFM

The possibility of generating only upper or lower sidebands provides some extra possibilities that AdFM originally did not. Figure 6 shows the steady state spectrum of a flute C4 tone. By using this sound as a source to single-sideband AdFM, we can produce a variety of spectral modifications. If set the index of modulation to 2 and the c:m ratio to 3, the resulting sound will be transposed downwards one octave and a fifth (F2). This is actually a perceived fundamental resulting from the interaction of the higher harmonics, as the sound does not contain any energy at that frequency (fig.7). The tone will still retain the gestural qualities of the original flute, but with a different timbral quality.



Figure 6. Flute C4 tone steady-state spectrum.

Similarly, an interesting effect is obtained by setting a non-integral *c:m* ratio, such as 2.4 (Fig.8). This will generate a sound that resembles multi-phonics, a technique of extended instrumental performance that generates an inharmonic mix of components.

By varying the index of modulation from 0 to its desired maximum value, transitions between 'normal' playing and multi-phonics-like tones can be achieved.



Figure 7. Steady-state spectrum of single-sideband AdFM using flute tone as input, with the modulation index I=2 and $c:m$ = 3.

Another interesting aspect of the technique is that, since the scaling of the generated components is not based on Bessel functions anymore, we have a different timbral evolution to straight AdFM. The effect of changing the index of modulation tends to create a 'moving formant', in addition to the increase in components. This region of spectral boost will rise in frequency proportionally to index increments. This effect gives the method a distinct sound, which cannot, however, be fine-tuned. The Asymmetric AdFM technique will allow for a better control of this feature.



Figure 8. Steady-state spectrum of single-sideband flute-based AdFM, with I=2 and c:m=2.45

4.2. Asymmetric AdFM

In Asymmetric AdFM we have, as discussed above, an extra parameter, *r*, which controls the symmetry of the spectrum. This can be used very much like a filter frequency control, but of course its effectiveness will ultimately depend on the index of modulation, as this controls the overall signal bandwidth. Both forms of the technique allow for the displacement of the centre of symmetry from the carrier to another frequency, allowing the control of formants. The second one, which exhibits modified Bessel functions in its expansion, can also make the formant shape more regular.



Figure 9. Bassoon C2 tone, steady-state spectrum

Using a low-frequency fundamental of a bassoon tone (C2), we can demonstrate the generation of formants in Asymmetric AdFM. Fig.9 shows the steady state spectrum of the original sound and in Fig.10 we see a 'symmetric' (ie. r=1) AdFM output. The spectrum shows a generally decreasing envelope, with just a few irregularities. Next, in Fig.11 we have the Bessel-based (original) method of Asymmetric AdFM, applied with the same index of modulation (1.5), but now with the *r* parameter set to 3. This displaces the symmetry of all of the modulated 'carriers' (ie. each componet of the original bassoon tone), resulting in a strong formant region around 2000 Hz.



Figure 10. 'Symmetric' AdFM spectrum using bassoon tone of fig.8, with I=1.5 (r = 1) and c:m = 1.



Figure 11. Asymmetric AdFM (original method) spectrum using bassoon tone of fig.8, with I=1.5, r=3 and c:m=1.



Figure 12. Asymmetric AdFM (modified Bessel-based method) spectrum using bassoon tone of fig.8, with I=1.5, r=3 and c:m=1.

Keeping the same parameters, but now using the second method (modified Bessel-based), we have a similar result, but perhaps with better defined formant regions (Fig.12). Moreover, here the shape of the formant region will be less distorted when the r and I (mod. index) parameters are changed, in comparison with the first method. By adjusting these, it is possible to displace the formant regions as required by a particular application. The extra flexibility provided by the technique allows for a significant enhancement of AdFM, with only a little extra computational cost.

One minor complication exists in relation to the normalisation factors $I_0(k[r-1/r])^{-1/2}$ in Eq.3 and $I_0(k[r+1/r])^{-1/2}$ in Eq.4. The most efficient way of implementing the logarithm of a modified Bessel function used in the formula is to use table lookup. In order to do so, we will need to estimate the maximum value that the lookup index will assume. If we under or over-estimate this value, the normalisation will either reduce the signal too much or have little effect. So, it is necessary to know, beforehand, the range of values of the modulation index and r parameters that will be employed, with quite some accuracy. If these ranges are allowed to vary, for each performance run a different table will need to be constructed.

4.3. Adaptive SpSB

The resulting spectra of ASpSB instruments are very similar to those of AdFM, as this technique is actually a refinement of the heterodyne adaptive FM design. This, in turn, has been demonstrated to be effectively a multicarrier process. Here, as it is possible to choose which one of the sideband groups we will use, we have a wider variety of possible effects.



Figure 13. Trumpet C4 tone, steady-state spectrum



Figure 14. Trumpet C4 tone, steady-state spectrum, processed through ASpSB, using upper sideband outputs only, with c:m =1:0.1 and I=2, generating a 'growl' sound.

Figures 13 and 14 demonstrate one interesting example using a trumpet tone as input, where ASpSB is used to add some nearby components to each harmonic, creating a 'growling' sound. For this example, we have set a c:m ratio of 1:0.1, an index of modulation of 2 and used only the upper sideband outputs (both even and odd). Two or three extra components are added to the upper side of each harmonic, generating the beating effects that characterise the 'growl'.

Another interesting example is the use of the lower sidebands to produce sub-harmonics, generating a change of timbre and of pitch. This effect is shown in the spectrogram on Fig.15, which shows an oboe G4 tone being slowly morphed into a lower-pitch sound by increasing the index of modulation, with *c:m* set to 1.5. This example only uses the lower sideband outputs.

The full range of heterodyne-based AdFM effects are available here, with the extra possibilities offered by the separate sideband outputs. Other interesting results might be obtained by placing the different sideband groups in separate spatial positions with multichannel audio. In addition, further processing might be added individually to the different outputs.



Figure 15. Oboe G4 tone with the slow introduction of sub-harmonics, through ASpSB with *c:m* = 3:2 . *I* is initially set at 0 and then increased from 0 to 5

5. CONCLUSION

In this work, we have shown three alternatives to the basic forms of Adaptive FM synthesis, which allow for the generation of asymmetric spectra. Two of the techniques are based on well known formulae, developed for standard FM synthesis and the third is a novel method, introduced by these authors. A number of examples have shown the potential applications for the techniques, demonstrating their specific strengths. In particular, the Split-Sideband method proves to be quite robust and flexible for the generation of a variety of effects. It is expected that the introduction of these methods will help develop renewed interest in the area of distortion synthesis with adaptive applications.

6. REFERENCES

[1] V. Lazzarini, J. Timoney and T. Lysaght, "Adaptive FM Synthesis". *Proceedings of the 10th International Conference on Digital Audio Effects*, Bordeaux, France, pp.21-26, 2007

[2] V. Lazzarini, J. Timoney and T. Lysaght, "The Generation of Natural-Synthetic Spectra by Means of Adaptive Frequency Modulation". *Computer Music Journal*, 32 (2), to appear, 2008

[3] J. Chowning, "The Synthesis of Complex Audio Spectra by Means of Frequency Modulation", *Journal of the Audio Engineering Society*, 21, pp. 526-34, 1973

[4] V. Verfaille, U. Zolzer and D. Arfib, "Adaptive Digital Audio Effects (a-DAFx): a new class of sound transformations", *IEEE Transactions on Speech and Audio Processing*, 14, (5), pp.1817- 1831, 2006

[5] J. A. Moorer, "The Synthesis of Complex Audio Spectra by Means of Discrete Summation Formulas", *Journal of the Audio Engineering Society*, 24 (9), pp.717-727, 1976

[6] J. A. Moorer, "Signal Processing Aspects of Computer Music: A Survey", *Proceedings of the IEEE*, 65 (8), pp.1108 - 1141,1977

[7] J. P. Palamin, P. Palamin and A. Ronveaux, "A Method of Generating and Controlling Musical Asymmetrical Spectra", *Journal of the Audio Engineering Society*, 36 (9), pp. 671-685, 1988

[8] V. Lazzarini, J. Timoney and T. Lysaght, "Split-Sideband Synthesis", *Proceedings of the ICMC 2008*, Queen's University Belfast, Northern Ireland, to appear, 2008

[9] S. Wardle, "A Hilbert Transform Frequency Shifter for Audio", *Proceedings of the International Conference on Digital Audio Effects*, Barcelona, Spain, pp.25-29, 1999

[10] F. W. Olver, "Bessel Functions of Integer Order", in: M. Abramowitz and I. A. Stegun, eds., *Handbook of Mathematical Functions*, Dover, New York, pp. 355-389, 1965

[11] J. ffitch, "On the Design of Csound5", *Proceedings of the 3rd Linux Audio Conference*, ZKM, Karlsruhe, Germany, pp. 37-42, 2005

[12] M. Le Brun, "Digital Waveshaping Synthesis", *Journal of the Audio Engineering Society*, 27(4), pp.250-266, 1979.

[13] G.N. Watson. *A Treatise on the Theory of Bessel Functions.* 2nd edition, Cambridge University Press, Cambridge, pp.22-23, 1944

[14] Cizek, V. "Discrete Hilbert Transform", *IEEE Transactions on Audio Electroacoustics*, 18 (4), pp340-343, 1970